Modeling Free Trade Areas in a World Trade Simulation Model

Thomas F. Rutherford

October 17, 2005

Calibrated equilibrium models are most commonly formulated as nonlinear systems of equations. This format is appealing to most economsts because it is familiar and straight-forward. In the system-of-equations setting the key logical consistency requirement is that the number of equations equals the number of variables. Provided these dimensions match up, the model is likely to work. Unlike complementarity-based models nonlinear equations need not be oriented, nor do variables need to be explicit matched up with equations.

It is quite common, however, to encounter modelling issues which call for complementarity methods when a model is used to address specific policy issues or intervention measures. This paper illustrates how the need for complementarity can arise when a "generic system of equations model" is extended to deal with a specific policy issue.

The policy issue I consider involves the formation of a Free Trade Area (FTA). In an FTA commodities are freely traded between regions within the FTA, but regions may also trade outside the area subject to tariff. This policy implies that imports of good i to countries within the FTA either from the global market or from the FTA market. The sourcing decision is based on relative profitability, and the model therefore portrays a canonical feature of complementarity problems.

1 Global Model Formulation

Consider a partial equilibrium model of world trade in which regional trade levels are determined by own-price elasticities of import demand and export supply:

- Import demand for good i in region r (M_{ir}), depending on the reference level of imports ($\bar{m}$), the world market price (p), the tariff rate (t), the reference import price ($\bar{p}$), the exchange rate (e) and the elasticity of import demand (ϵ:

$$M_{ir} = \bar{m}_{ir} \left(\frac{p_i(1+t_{ir})}{\bar{p}_{ir} e_r} \right)^{\epsilon_{ir}} \tag{1}$$

- Export supply for good i from region r (X_{ir}), depending on the reference level of exports ($\bar{x}$), the world market price (p), the exchange rate (e) and the elasticity of export supply (η):

$$X_{ir} = \bar{x}_{ir} \left(\frac{p_i}{e_r} \right)^{\eta_{ir}} \tag{2}$$

- World market clearance for good i which equates aggregate regional exports and regional imports:

$$\sum_r X_{ir} = \sum_r M_{ir} \tag{3}$$

- Current account balance for region r relating the world market value of imports (pM), exports (pX) and the exogenously-specified current account balance (b):

$$\sum_r p_i \left(M_{ir} - X_{ir} \right) = b_r \tag{4}$$

An equilibrium is defined as a set of variables (M, X, p, and e) which solve equations (1) through (4).

The benchmark data are balanced with equilibrium market prices equal to unity for all goods when:

$$\sum_r \bar{m}_{ir} = \sum_r x_{ir}$$

and

$$\bar{p}_{ir} = 1 + t_{ir}$$

and

$$b_r = \sum_i \bar{m}_{ir} - \bar{x}_{ir}$$

2 Model Implementation

Here is GAMS code for a specific instance of this model:

```
$Title A Vinerian model of World Trade

*       Read the XL workbook if the GDX data file is not found:

$onecho >gdxxrw.txt
set=i   rng=sets!i2 rdim=1 cdim=0
set=r   rng=sets!b2 rdim=1 cdim=0
par=eta rng=eta!b2
par=epsilon rng=epsilon!b2
par=t rng=t!b2
```

```
par=xs0 rng=xs0!b2
par=md0 rng=md0!b2
$offecho

$if not exist tradedata.gdx $call gdxxrw i=tradedata.xls o=tradedata.gdx @gdxxrw.txt

$gdxin 'tradedata.gdx'

set i(*) Commodities,
r(*) Regions;

$load i r

parameter
        md0(i,r)        Base year import demand,
        xs0(i,r)        Base year exports,
        t(i,r)          Tariff rate,
        epsilon(i,r)      Import demand elasticity,
        eta(i,r)        Export supply elasticity;

$load md0 xs0 t epsilon eta

parameter
        b(r)            Current account deficit
        pm0(i,r)        Reference price of imports,
        xsd(i)          Excess demand;

b(r) = sum(i, md0(i,r)-xs0(i,r));
xsd(i) = sum(r, md0(i,r) - xs0(i,r));
display xsd;

pm0(i,r) = 1 + t(i,r);

variables
        M(i,r)  Imports
        X(i,r)  Exports
        P(i)    World market price
        E(r)    Exchange rate in region r;


equations       mdemand, xsupply, market, caccount;

mdemand(i,r)..  M(i,r) =e= md0(i,r) * (P(i)*(1+t(i,r))/(E(r)*pm0(i,r)))**epsilon(i,r);

xsupply(i,r)..  X(i,r) =e= xs0(i,r) * (P(i)/E(r))**eta(i,r);

market(i)..     sum(r, M(i,r) - X(i,r)) =e= 0;

caccount(r)..   sum(i, P(i) * (X(i,r) - M(i,r))) + b(r) =e= 0;

model wtsm /mdemand.M, xsupply.X, market.P, caccount.E/;
```

```
M.l(i,r) = md0(i,r);   X.l(i,r) = xs0(i,r);  P.l(i)   = 1;  E.l(r)   = 1;

wtsm.iterlim = 0;
solve wtsm using mcp;
wtsm.iterlim = 10000;
```

3 Sample Calculation

Replication of the benchmark dataset is not particularly interesting. To show how this model can be used to do some economic analysis, let us evaluate the economic impact of unilateral free trade policies using this specific dataset. An evaluation of the impact of policies requires that we calculate indices describing policy outcomes. Social surplus in this model is one such index, and it can be calculated on the basis of three components:

1. Tariff revenue:
$$\Delta TR_r = \sum_i \left(t_{ir} \frac{p_i M_{ir}}{e_r} - (\bar{p}_{ir} - 1)\bar{m}_{ir} \right)$$
2. Consumer surplus:
$$\Delta CS_r = \sum_i \left(\bar{p}_{ir} - \frac{p_{ir}(1+t_{ir})}{e_r} \right) \frac{\bar{m}_{ir} + M_{ir}}{2}$$
3. Producer surplus:
$$\Delta PS_r = \sum_i \left(\frac{p_{ir}}{e_r} - 1 \right) \frac{\bar{x}_{ir} + X_{ir}}{2}$$

The following GAMS code evaluates the welfare effect of unilateral liberalization policies for each of the countries and regions in this dataset:

```
wtsm.iterlim = 10000;

parameter       surplus         Changes in various components of social surplus;

alias (rfta,r);
loop(rfta,
        t(i,r) = pm0(i,r) - 1;
        t(i,rfta) = 0;

        solve wtsm using mcp;

        surplus(rfta,r,"Tariff") =
                sum(i, P.L(i)*t(i,r)*M.L(i,r)/E.L(r) - (pm0(i,r)-1)*md0(i,r));
        surplus(rfta,r,"consumer") =
                sum(i, (pm0(i,r)-P.L(i)*(1+t(i,r))/E.L(r))*(md0(i,r)+M.L(i,r))/2);
        surplus(rfta,r,"producer") =
```

```
                sum(i, (P.L(i)/E.L(r)-1) * (xs0(i,r)+X.L(i,r))/2);
);

parameter       summary         Summary of the welfare impact of unilateral free trade;

set s /Tariff, Consumer, Producer/;

summary(r,"Region","Tariff") = surplus(r,r,"Tariff");
summary(r,"Region","Consumer") = surplus(r,r,"consumer");
summary(r,"Region","Producer") = surplus(r,r,"producer");
summary(r,"Region","Total") = sum(s, summary(r,"Region",s));

alias (r,rr);

summary(r,"World","Consumer") = sum(rr,surplus(r,rr,"consumer"));
summary(r,"World","Producer") = sum(rr,surplus(r,rr,"producer"));
summary(r,"World","Tariff")   = sum(rr,surplus(r,rr,"Tariff"));
summary(r,"World","Total") = sum(s, summary(r,"World",s));
option summary:1:1:2;
display summary;
```

Table 1: Regional Aggregation for Illustrative Dataset

ALB	Albania,
BIH	Bosnia -Herzegovina,
BGR	Bulgaria,
HRV	Croatia,
MKD	TFYR Macedonia,
MDA	Rep. of Moldava,
ROM	Romania,
YUG	Serbia - Montenegro,
TUR	Turkey,
UKR	Ukraine,
EUO	EU15 - Original 15 EU,
EUN	EU10 - New 10 EU,
BLR	Belarus,
RUS	Russia,
CHE	Switzerland,
USA	USA,
JPN	Japan,
CHN	China,
DCO	Other Developed Countries,
EMP	Euro-Mediterranean Partnership Countries,
WAO	Other West Asia,
AFO	Other Africa,
ASC	Central Asia - Fm. USSR-Asia,
ASO	Other Asia,
AMO	Other America,
ROW	Rest of World (Other Developing Countries)

Table 2: Unilateral Liberalization in the World Trade Simulation Model

	Regional Welfare				Global Welfare			
	Tariff	Consumer	Producer	Total	Tariff	Consumer	Producer	Total
ALB	-0.2		0.1	-0.1	-0.2	-0.2	-0.4	-0.8
BIH	-0.4	0.1	0.1	-0.2	-0.4	-0.2	-0.3	-0.9
BGR	-1.1	0.3	0.6	-0.2	-1.1	0.1	0.2	-0.8
HRV	-1.5	0.3	0.6	-0.6	-1.6	0.1	0.2	-1.3
MKD	-0.2	0.1	0.1		-0.2	-0.1	-0.2	-0.5
MDA	-0.2	0.1	0.1		-0.2	-0.2	-0.3	-0.7
ROM	-2.7	0.9	1.5	-0.3	-2.7	0.7	1.2	-0.8
YUG	-0.7	0.1	0.2	-0.4	-0.7	-0.1	-0.2	-1.0
TUR	-6.9	2.0	3.8	-1.1	-6.9	1.9	3.3	-1.7
UKR	-1.6	0.8	1.3	0.5	-1.6	0.6	1.0	
EUO	-88.9	35.7	59.2	6.0	-88.9	34.6	59.8	5.5
EUN	-24.5	8.5	15.0	-1.0	-24.6	8.2	14.7	-1.7
BLR	-1.1	0.5	0.8	0.2	-1.1	0.4	0.5	-0.2
RUS	-6.8	3.8	7.8	4.8	-6.8	4.5	7.0	4.7
CHE	-11.0	4.8	8.0	1.8	-11.0	4.5	7.8	1.3
USA	-35.3	11.0	15.0	-9.3	-35.3	9.8	16.0	-9.5
JPN	-10.8	6.3	6.5	2.0	-10.7	4.9	7.9	2.1
CHN	-41.4	18.2	35.7	12.5	-41.4	19.3	35.1	13.0
DCO	-18.0	6.2	13.8	2.0	-18.1	7.4	12.7	2.0
EMP	-12.9	4.4	8.8	0.3	-12.9	4.9	7.9	-0.1
WAO	-14.3	5.3	15.9	6.9	-14.4	8.4	13.6	7.6
AFO	-10.0	4.0	8.6	2.6	-10.1	4.7	7.6	2.2
ASC	-2.1	0.8	1.8	0.5	-2.1	0.8	1.3	
ASO	-108.0	39.7	74.9	6.6	-108.1	40.0	74.2	6.1
AMO	-37.3	14.6	28.5	5.8	-37.5	16.3	26.7	5.5
ROW	-48.7	4.1	12.1	-32.5	-48.8	4.9	10.3	-33.6

4 Economic Equilibrium with a Free Trade Area

When a Free Trade Area (FTA) is created, tariffs on imports from within the FTA are set to zero while tariffs on imports from other regions may be differentiated. (In the present example, it will be assumed that regional tariffs on non-FTA trade flows are unchanged.) This implies that there will be two market prices for each commodity, one which describes the price of goods traded on the international market, and another which is the price of goods traded within the FTA. Relative profitability determines whether regions with the FTA choose to import from other FTA countries.

Equilibrium conditions are then revised as follows:

- The import demand market for good i in region r equates the supply of imports from outside the FTA (M_{ir}) and inside the FTA (M_{ir}^{FTA}) with the demand for imports:

$$M_{ir} + M_{ir}^{FTA} = \bar{m}_{ir} \left(\frac{pm_{ir}}{\bar{p}_{ir} e_r} \right)^{\epsilon_{ir}} \tag{5}$$

  in which only regions within the FTA are permitted to import from that source:

$$M_{ir}^{FTA} = 0 \quad \forall r \notin FTA$$

- Exports of good i from region r consiste of those destine outside the FTA (X_{ir}) and those sent to other FTA regions:

$$X_{ir} + X_{ir}^{FTA} = \bar{x}_{ir} \left(\frac{px_{ir}}{e_r} \right)^{\eta_{ir}} \tag{6}$$

  while only regions within the FTA are permitted to export to that market:

$$X_{ir}^{FTA} = 0 \quad \forall r \notin FTA$$

- World market clearance for good i

$$\sum_r X_{ir} = \sum_r M_{ir} \tag{7}$$

- FTA market clearance for good i

$$\sum_{r \in FTA} X_{ir}^{FTA} = \sum_{r \in FTA} M_{ir}^{FTA} \tag{8}$$

- Complementary slackness conditions characterize trade flows:

$$\begin{array}{lcrr} M_{ir}^{FTA} \geq 0 & \perp & p_i^{FTA} \geq pm_{ir} & \forall r \in FTA \\ X_{ir}^{FTA} \geq 0 & \perp & px_{ir} \geq p_i^{FTA} & \forall r \in FTA \\ M_{ir} \geq 0 & \perp & p_i(1+t_{ir}) \geq pm_{ir} & \forall r \\ X_{ir} \geq 0 & \perp & px_{ir} \geq p_i & \forall r \end{array} \tag{9}$$

- Current account balance for region r relating the aggregate value of imports less exports to the exogenously-specified current account balance (b):

$$\begin{array}{lll} \sum_i p_i \left(M_{ir} - X_{ir}\right) & = b_r & r \notin FTA \\ \sum_i p_i \left(M_{ir} - X_{ir}\right) + \sum_i p_i^{FTA} \left(M_{ir}^{FTA} + X_{ir}^{FTA}\right) & = b_r & r \in FTA \end{array} \tag{10}$$

```
positive
variables
        M(i,r)          Imports
        X(i,r)          Exports
        MFTA(i,r)       Imports from the Free Trade Area
        XFTA(i,r)       Exports to  the Free Trade Area
        P(i)            World market price
        PFTA(i)         Market price within the FTA
        PM(i,r)         Import price
        PX(i,r)         Export price
        E(r)            Regional exchange rate;

equations       mdemand, meq, mftaeq, xsupply, xeq, xftaeq, ftamarket, market, caccount;

mdemand(i,r)..

        M(i,r) + MFTA(i,r)$fta(r) =e= md0(i,r) * (PM(i,r)/(E(r)*pm0(i,r)))**epsilon(i,r);

meq(i,r)..

        P(i) * (1+t(i,r)) =G= PM(i,r);

mftaeq(i,r)$fta(r)..

        PFTA(i) =G= PM(i,r);

xsupply(i,r)..

        X(i,r) + XFTA(i,r)$fta(r) =e= xs0(i,r) * (PX(i,r) /E(r))**eta(i,r);

xeq(i,r)..

        PX(i,r) =G= P(i);

xftaeq(i,r)$fta(r)..

        PX(i,r) =G= PFTA(i);

ftamarket(i)$card(fta)..

        sum(fta(r), MFTA(i,r) - XFTA(i,r)) =E= 0;

market(i)..
```

```
        lsum(r, M(i,r) - X(i,r)) =E= 0;

caccount(r)..

        sum(i,  P(i)     * (X(i,r)    - M(i,r)) +
        (PFTA(i) * (XFTA(i,r) - MFTA(i,r)))$fta(r)) + b(r) =e= 0;

model global /  mdemand.PM, meq.M, mftaeq.MFTA, xsupply.PX,
                xeq.X, xftaeq.XFTA, ftamarket.PFTA, market.P, caccount.E/;

M.l(i,r) = md0(i,r);
X.l(i,r) = xs0(i,r);
P.l(i)   = 1;
PM.L(i,r) = pm0(i,r);
PX.L(i,r) = 1;
E.l(r)   = 1;
PFTA.l(i) = 1;

*       Replicate the benchmark in which there is no FTA:

fta(r) = no;
nfta(r) = (not fta(r));

global.iterlim = 0;
solve global using mcp;

*       Insert some bounds to avoid bad function calls:

E.LO(r) = 0.001;
P.LO(i) = 0.001;
PFTA.LO(i) = 0.001;
PM.LO(i,r) = 0.001;
PX.LO(i,r) = 0.001;

*       Define member states in the FTA:

set     k(r)    States creating the FTA
        /ALB, BIH, BGR, HRV, MKD, MDA, ROM, YUG, TUR, UKR, EUO /;

fta(k) = yes;
nfta(r) = (not fta(r));

global.iterlim = 10000;
solve global using mcp;
```

This type of model involves a large number of explicit complementarity conditions, as the calculation of the new equilibrium involves determination of the trade pattern as well as the equilibrium prices. This characteristic of the model formulation is evidenced in PATH's iteration log which

reports a large number of minor iterations in the first two major iterations. During these two iterations, the solution algorithm is sorting out which regions and goods are traded in the FTA markets. After having identified the set of import and export activities which are operated at positive intensity, the corresponding square system of nonlinear equations is solved in major iterations 2 to 12, with a diminishing number of basis adjustements in each iteration:

```
Major Iteration Log
major minor  func  grad  residual    step  type prox    inorm  (label)
    0     0    37     4 3.6924e+000           I 3.0e-002 1.6e+000    (caccount(EU0))

Minor Iteration Log
minor      t          z     w     v   art ckpts  enter    leave
  500 -4.4762e-002  2400   439     0     0     9 w[ 1570] z[ 1558]
 1000  8.4196e-004  2788    51     0     0    19 w[  344] z[  473]
 1500 -1.2679e-001  2389   450     0     0    29 z[ 1431] w[ 1436]
 2000  5.1532e-004  2795    44     0     0    39 z[  472] w[  340]
 2500 -4.1567e-002  2409   430     0     0    49 z[ 1440] w[ 1450]
 3000  4.9804e-003  2744    95     0     0    58 w[  997] z[ 1001]
    1    32    50     5 7.7193e+000  4.4e-005 RB 1.2e-002 5.2e+000 (mdemand(ma,EU0))

Minor Iteration Log
minor      t          z     w     v   art ckpts  enter    leave
  500 -1.4724e-002  2430   409     0     0     9 z[  778] w[  602]
 1000 -1.1928e-002  2731   108     0     0    19 w[ 1416] z[ 1412]
 1500 -1.3517e-002  2532   307     0     0    29 z[  710] w[ 1455]
 2000 -1.2074e-002  2629   210     0     0    39 w[  679] z[  675]
 2500 -1.2069e-002  2634   205     0     0    49 z[  671] w[ 1424]
 3000 -1.3832e-002  2527   312     0     0    58 w[ 1456] z[  572]
 3500 -1.1924e-002  2736   103     0     0    68 z[ 1469] w[ 1600]
 4000 -1.5080e-002  2425   414     0     0    78 w[  655] z[  723]

Major Iteration Log
major minor  func  grad  residual    step  type prox    inorm  (label)
    2  4494    51     6 7.7191e+000  1.0e+000 CO 4.8e-003 5.2e+000 (mdemand(ma,EU0))
    3   444    52     7 1.3521e+000  1.0e+000 SO 1.9e-003 6.7e-001 (xsupply(ma,EU0))
    4    33    53     8 3.1884e-001  1.0e+000 SO 7.7e-004 2.5e-001 (caccount(EU0))
    5    29    54     9 1.7391e-001  1.0e+000 SO 3.1e-004 1.3e-001 (mdemand(fs,EU0))
    6    12    55    10 2.0984e-001  1.0e+000 SO 1.2e-004 1.7e-001 (caccount(EU0))
    7    34    56    11 1.0812e-001  1.0e+000 SO 4.9e-005 6.9e-002 (mdemand(rp,EU0))
    8    49    57    12 1.7921e-001  1.0e+000 SO 2.0e-005 1.4e-001 (mdemand(ch,EU0))
    9    51    58    13 1.3230e-001  1.0e+000 SO 7.9e-006 1.2e-001 (mdemand(ma,EU0))
   10    19    59    14 1.0936e-001  1.0e+000 SO 3.2e-006 6.4e-002 (ftamarket(te))
   11     2    60    15 1.4373e-006  1.0e+000 SO 1.3e-006 1.1e-006 (mdemand(ma,EU0))
   12     1    61    16 6.3698e-011  1.0e+000 SO 1.4e-007 3.8e-011 (mdemand(ma,EU0))
```

We finally summarize the economic effects of the FTA. The GAMS code for these results are as follows:

```
parameter ssummary Sectoral Results Summary (\% change),
rsummary Regional Results Summary (\% change);

ssummary(i,"P") = 100 * (P.l(i)-1);
ssummary(i,"PFTA") = 100 * (PFTA.l(i)-1);
ssummary(i,"Trade") = 100 *(sum(r, M.l(i,r)+MFTA.L(i,r))/sum(r, md0(i,r))-1);
rsummary(r,"E") = 100 * (E.L(r)-1);
rsummary(r,"M") = 100 * (sum(i, M.L(i,r)+MFTA.L(i,r))/sum(i,md0(i,r))-1);
rsummary(r,"X") = 100 * (sum(i, X.L(i,r)+XFTA.L(i,r))/sum(i,xs0(i,r))-1);
display ssummary,rsummary;

parameter surplus Social surplus impacts of FTA;
surplus(r,"Tariff") = round(sum(i, P.L(i)*t(i,r)*M.L(i,r)/E.L(r) - (pm0(i,r)-1)*md0(i,r)),1);
surplus(r,"Consumer") = sum(i, (pm0(i,r)-PM.L(i,r)/E.L(r)) * (md0(i,r)+M.L(i,r))/2);
surplus(r,"Producer") = sum(i, (PX.L(i,r)/E.L(r)-1)        * (xs0(i,r)+X.L(i,r))/2);

set s /Tariff,Consumer,Producer/;

surplus(r,"Total") = sum(s, surplus(r,s));
surplus("FTA",s) = sum(fta, surplus(fta,s));
surplus("nFTA",s) = sum(nfta, surplus(nfta,s));
surplus("FTA","Total") = sum(s,surplus("FTA",s));
surplus("nFTA","Total") = sum(s, surplus("NFTA",s));
option surplus:1;
display surplus;
```

The following lines generate output in the form an xls worksheet, `results.xls`:

```
$onecho >gdxxrw.txt
par=ssummary rng=FTA!B5 merge
par=rsummary rng=FTA!B36 merge
par=surplus rng=FTA!B71 merge'
$offecho

execute_unload 'ftamodel.gdx', ssummary, rsummary, surplus;
execute 'gdxxrw i=ftamodel.gdx o=results.xls @gdxxrw.txt
```

Table 3: Sectoral Results Summary (% change)

	P	PFTA	Trade
Live animals & animal products	-0.9	3.5	0.8
Vegetable products	-0.9	4.1	0.7
Fats & oils	1.1	1.1	0.9
Manufactured foodstuffs	3.3	3.3	2.7
Mineral products	0.2	0.4	0.0
Chemical	1.0	1.0	1.2
Rubber & plastics	1.1	1.1	1.2
Hides & leather products	0.6	0.6	0.7
Cork & wood articles	-0.4	2.4	0.6
Pulp & paper products	0.2	0.2	0.4
Textiles & apparel	-0.7	5.8	1.7
Footwear & other made-up articles	-0.6	4.6	1.4
Stone & mineral products	0.9	0.9	1.2
Precious stones & jewellery	0.2	0.9	0.3
Base metals & metal products	0.7	0.7	0.9
Machinery	0.6	0.6	0.5
Transport equipment	0.7	0.7	0.9
Professional equipment	0.8	0.8	0.7
Arms & ammunition	0.5	0.5	0.6
Miscellaneous manufactures	-0.3	2.4	0.6
Works of art	0.4	0.4	0.6
Other goods	-0.2	-0.2	0.0

Table 4: Regional Results Summary (% change)

	E	M	X
Albania	-7.8	2.4	11.7
Bosnia -Herzegovina	-7.1	3.4	9.1
Bulgaria	-4.8	5.6	7.1
Croatia	-6.8	3.3	8.4
TFYR Macedonia	-4.6	5.1	7.3
Rep. of Moldava	-4.9	6.5	7.1
Romania	-5.0	6.9	7.8
Serbia - Montenegro	-6.5	3.0	8.3
Turkey	-5.0	6.5	7.7
Ukraine	-3.7	6.8	4.5
EU15 - Original 15 EU	-0.8	2.1	1.9
EU10 - New 10 EU	0.5	-0.1	0.0
Belarus	0.4	-0.4	-0.1
Russia	0.5	-0.3	-0.1
Switzerland	0.7	0.2	0.0
USA	0.5	0.0	0.1
Japan	0.7	0.4	-0.1
China	0.4	-0.3	-0.1
Other Developed Countries	0.5	-0.2	-0.1
Euro-Mediterranean Partnership Countries	0.4	-0.3	-0.1
Other West Asia	0.4	-0.2	-0.1
Other Africa	0.5	-0.1	-0.1
Central Asia - Fm. USSR-Asia	0.4	-0.4	-0.1
Other Asia	0.5	-0.1	0.0
Other America	0.5	-0.1	-0.1
Rest of World (Other Developing Countries)	0.3	-0.3	0.0

Table 5: Social surplus impacts of FTA

	Tariff	Consumer	Producer	Total
Albania	-0.2	0.0	0.0	-0.1
Bosnia -Herzegovina	-0.4	0.0	0.1	-0.3
Bulgaria	-1.1	0.1	0.4	-0.6
Croatia	-1.5	0.1	0.4	-1.0
TFYR Macedonia	-0.2	0.0	0.1	-0.1
Rep. of Moldava	-0.2	0.0	0.1	-0.1
Romania	-2.7	0.5	0.8	-1.4
Serbia - Montenegro	-0.7	0.0	0.2	-0.5
Turkey	-6.9	1.1	2.1	-3.7
Ukraine	-1.6	0.4	0.6	-0.6
EU15 - Original 15 EU	-87.7	14.6	32.6	-40.5
EU10 - New 10 EU	0.0	-0.1	-0.1	-0.2
Belarus	0.0	0.0	0.0	0.0
Russia	0.0	-0.1	-0.3	-0.4
Switzerland	0.0	0.1	0.0	0.1
USA	0.0	0.6	0.7	1.3
Japan	0.0	1.4	-0.4	0.9
China	-0.1	-0.5	-0.9	-1.5
Other Developed Countries	0.0	-0.4	-0.6	-0.9
Euro-Mediterranean Partnership Countries	0.0	-0.1	-0.1	-0.2
Other West Asia	0.0	-0.1	-0.5	-0.6
Other Africa	0.0	0.0	-0.2	-0.2
Central Asia - Fm. USSR-Asia	0.0	0.0	0.0	-0.1
Other Asia	-0.1	0.3	-0.6	-0.4
Other America	0.0	0.0	-0.4	-0.4
Rest of World (Other Developing Countries)	-0.1	-0.5	0.0	-0.6
Aggregate within FTA	-103.2	17.0	37.3	-48.9
Aggregate Outside FTA	-0.3	0.5	-3.5	-3.2